



# 4D Progress


4D incluye un componente integrado llamado 4D Progress. Este componente les permite abrir una o más barras de progreso en una misma ventana (como la interfaz del Finder en Mac OS).

Cada barra de progreso recibe un ID que es generado automáticamente por el método **Progress New**. Este ID es utilizado por todos los métodos proyecto del componente para administrar las propiedades y valores en las cajas de diálogo de progreso.























Los métodos proyecto para este componente se listan en la página **Métodos componente** del Explorador:



 Barras de progreso

 Lista alfabética de los comandos

# **Barras de progreso**

-  Progress Get Button Enabled
-  Progress Get Button Title
-  Progress Get Icon
-  Progress Get Message
-  Progress Get On Error Method
-  Progress Get On Stop Method
-  Progress Get Progress
-  Progress Get Title
-  Progress New
-  Progress QUIT
-  Progress SET BUTTON ENABLED
-  Progress SET BUTTON TITLE
-  Progress SET FONT SIZES
-  Progress SET FONTS
-  Progress SET ICON
-  Progress SET MESSAGE
-  Progress SET ON ERROR METHOD
-  Progress SET ON STOP METHOD
-  Progress SET PROGRESS
-  Progress SET TITLE
-  Progress SET WINDOW VISIBLE
-  Progress Stopped

## ⚙️ Progress Get Button Enabled

Progress Get Button Enabled ( id ) -> Resultado

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
Resultado	Booleano	↩	True = botón Stop mostrado; de lo contrario, False

### Descripción

---

El método **Progress Get Button Enabled** devuelve **True** cuando la barra de progreso designada por el parámetro *id* muestra un botón **Stop**. Si no muestra uno (funcionamiento por defecto), el método devuelve **False**.

## Progress Get Button Title

Progress Get Button Title ( id ) -> Resultado

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
Resultado	Texto	↩	Title of Stop button

### Descripción

---

**Nota:** este método sólo puede ser utilizado bajo Windows. Bajo Mac OS, los botones para detener no tienen títulos.

El método **Progress Get Button Title** devuelve el título actual del botón stop para la barra de progreso designada por el parámetro *id*.

Por defecto, el título es "Stop". Note que el método devuelve el título actual incluso si el botón **Stop** no se muestra.

## Progress Get Icon

Progress Get Icon ( id ) -> Resultado

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
Resultado	Imagen	↩	Icono de la barra de progreso

### Descripción

---

El método **Progress Get Icon** devuelve el icono de la barra de progreso designada por el parámetro *id*.

## Progress Get Message

Progress Get Message ( id ) -> Resultado

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
Resultado	Texto	↩	Mensaje de la barra de progreso

### **Descripción**

---

El método **Progress Get Message** devuelve el mensaje de la barra de progreso designada por el parámetro *id*.

## Progress Get On Error Method

Progress Get On Error Method -> Resultado

Parámetro	Tipo	Descripción
Resultado	Texto	 Nombre del método llamado en caso de error (si se define)

### Descripción

---

El método **Progress Get On Error Method** devuelve el nombre del método proyecto de la base local llamada en caso de error durante el uso de una barra de progreso.

Si no se define un método de error, este método devuelve una cadena vacía.

## Progress Get On Stop Method

Progress Get On Stop Method ( id ) -> Resultado

Parámetro	Tipo	Descripción
id	Entero largo	→ ID de la barra de progreso
Resultado	Texto	↪ Nombre del método llamado en caso de clic en el botón Stop (si se define)

### Descripción

---

El método **Progress Get On Stop Method** devuelve el nombre del método de proyecto de la base local llamada cuando el usuario hace clic en el botón **Stop** de la barra de progreso designada por el parámetro *id*.

Si ningún método está asociado al botón **Stop**, este método devuelve una cadena vacía.



## Progress Get Progress

Progress Get Progress ( id ) -> Resultado

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
Resultado	Real	↩	Valor de la barra de progreso

### Descripción

---

El método **Progress Get Progress** devuelve el valor actual asociado a la barra de progreso designada por el parámetro *id*.

## Progress Get Title

Progress Get Title ( id ) -> Resultado

<b>Parámetro</b>	<b>Tipo</b>		<b>Descripción</b>
id	Entero largo	→	ID de la barra de progreso
Resultado	Texto	↩	Título de la barra de progreso

### **Descripción**

---

El método **Progress Get Title** devuelve el título principal de la barra de progreso designada por el parámetro *id*.

## Progress New

Progress New -> Resultado

Parámetro	Tipo	Descripción
Resultado	Entero largo	 ID de la nueva barra de progreso

### Descripción

---

El método **Progress New** crea una nueva barra de progreso y devuelve su número de ID. Este número es único durante la duración de la barra de progreso pero puede ser reutilizado posteriormente.


La primera vez que se llama este método, se crea un proceso local y una nueva ventana centrada se abre sobre la ventana principal. Por defecto, esta ventana:

- contiene una barra de progreso indefinida
- no tiene un mensaje.

Mac OS



Windows



## Progress QUIT

Progress QUIT ( id )

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso

### Descripción

---

El método **Progress QUIT** cierra la barra de progreso referenciada por el parámetro *id*.

Si *id* designa la única barra de progreso mostrada, la ventana de progreso también se cierra (como también el proceso local). De lo contrario, la ventana se redimensiona con el fin de sólo contener las barras de progreso abiertas.

Puede pasar 0 en el parámetro *id* con el fin de detener todas las barras de progreso y cerrar la ventana de progreso.


### Ejemplo

---


Si la barra llamada "Copying folder 3" tiene el número de ID 3:

*Progress QUIT(3)*

Before




After




Si una ventana de progreso ya está abierta cuando se llama este método, esta ventana se redimensiona para mostrar una nueva barra de progreso debajo de la anterior en el mismo proceso:

## Mac OS



## Windows



## ⚙️ Progress SET BUTTON ENABLED

Progress SET BUTTON ENABLED ( id ; boton )



Parámetro	Tipo		Descripción
id	Entero largo	➔	ID de la barra de progreso
boton	Booleano	➔	True = Mostrar, False = Ocultar

### Descripción

---

El método **Progress SET BUTTON ENABLED** permite añadir un botón **Stop** a la barra de progreso designada por el parámetro *id*.

Por defecto, las barras de progreso no tienen botones **Stop**. Si pasa **True** en el parámetro *boton*, se muestra un botón:



Puede manejar el efecto del clic en el botón **Stop** utilizando el método **Progress SET ON STOP METHOD** o probando el valor del método **Progress Stopped**.

## ⚙️ Progress SET BUTTON TITLE

Progress SET BUTTON TITLE ( id ; titulo )


Parámetro	Tipo		Descripción
id	Entero largo	➡	ID de la barra de progreso
titulo	Texto	➡	Título del botón Stop (Windows)

### Descripción

---

**Nota:** este método sólo puede utilizarse bajo Windows. Bajo Mac OS, los botones stop no tiene títulos.

El método **Progress SET BUTTON TITLE** permite modificar el título del botón **Stop** de la barra de progreso designada poner el parámetro *id*. Por defecto, el título de este botón es "Stop":




Por defecto, las barras de progreso no tienen botones **Stop**. Para que una barra de progresión contenga un botón Stop, debe utilizar el método **Progress SET BUTTON ENABLED**.

### Ejemplo

---

Usted quiere utilizar el título "Abort":

```
<>ID:=Progress New  
...  
Progress SET BUTTON TITLE(<>ID;"Abort")
```



## ⚙️ Progress SET FONT SIZES

Progress SET FONT SIZES ( tamTitulos {; tamMensajes {; tamBotones}} )

Parámetro	Tipo	Descripción
tamTitulos	Entero largo →	Tamaño de la fuente para los títulos
tamMensajes	Entero largo →	Tamaño de la fuente para los mensajes
tamBotones	Entero largo →	(Windows) Tamaño de la fuente para los botones Stop

### Descripción

---

El método **Progress SET FONT SIZES** permite cambiar el tamaño de la fuente utilizada para los diferentes textos mostrados en todas las ventanas de progreso. En los parámetros *tamTitulos*, *tamMensajes* y *tamBotones*, pase los tamaños de las fuentes a utilizar. Si no quiere modificar un tamaño, pase -1 en el parámetro correspondiente.

### Ejemplo 1

---

Queremos cambiar únicamente el tamaño de los mensajes:


```
Progress SET FONT SIZES(-1;13)
```

### Ejemplo 2

---

Queremos cambiar el tamaño de los títulos y de los mensajes:

```
Progress SET FONTS("Arial Black";"Arial narrow")  
Progress SET FONT SIZES(13;12)
```





## Progress SET FONTS

Progress SET FONTS ( fuenteTitulos {; fuenteMensajes {; fuenteBotones}} )

Parámetro	Tipo	Descripción
fuenteTitulos	Texto →	Fuente a utilizar para los títulos
fuenteMensajes	Texto →	Fuente a utilizar para los mensajes
fuenteBotones	Texto →	(Windows) Fuente a utilizar para los botones Stop

### Descripción

---

El método **Progress SET FONTS** permite cambiar las fuentes utilizadas para los diferentes textos mostrados en todas las ventanas de progreso.

En los parámetros *fuenteTitulos*, *fuenteMensajes* y *fuenteBotones*, pase los nombres de las fuentes a utilizar. Si no desea modificar una fuente, pase una cadena vacía ("") en el parámetro correspondiente.

### Ejemplo

---

Queremos cambiar la fuente de los mensajes:

```
Progress SET FONTS("";"Arial")
```

## ⚙️ Progress SET ICON

Progress SET ICON ( id ; icono {; primerPlano} )

Parámetro	Tipo		Descripción
id	Entero largo	➔	ID de la barra de progreso
icono	Imagen	➔	Imagen a mostrar como icono
primerPlano	Booleano	➔	Mostrar la barra de progreso en el primer plano

### Descripción

---

El método **Progress SET ICON** permite modificar el icono mostrado en la barra de progreso. Por defecto, se muestran los siguientes iconos:



Windows



Mac OS

En *id*, pase el número de id único de la barra de progreso, devuelto por el método **Progress New**.


En *icono*, pase la imagen (variable o campo) a utilizar como icono en la ventana de la barra de progreso. El tamaño máximo de este icono debe ser:

- bajo Mac OS, 40 x 40 píxeles
- bajo Windows, 40 x 80 píxeles


Si pasa un icono con un tamaño que es menor a estos límites, se centra y no se redimensiona. Sin embargo, cuando su tamaño supera estos límites, se centra y redimensiona.

Pase **True** en *primerPlano* si quiere forzar el paso de la ventana de progreso al primer plano de la aplicación.

Ejemplos de iconos personalizados:



Windows



Mac OS

## Progress SET MESSAGE

Progress SET MESSAGE ( id ; mensaje {; primerPlano} )

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
mensaje	Texto	→	Mensaje de la barra de progreso
primerPlano	Booleano	→	Mostrar la barra de progreso en el primer plano

### Descripción

---

El método **Progress SET MESSAGE** permite modificar el mensaje mostrado en la barra de progreso.

En *id*, pase el número de ID único de la barra de progreso, devuelto por el método **Progress New**.

En *mensaje*, pase el texto a modificar bajo el título principal (Windows) o bajo la barra de progreso (Mac OS).

Pase **True** en *primerPlano* si quiere forzar el paso de la ventana de progreso al primer plano de la aplicación.

## ⚙️ Progress SET ON ERROR METHOD

Progress SET ON ERROR METHOD ( *nomMetodo* )

Parámetro	Tipo	Descripción
<i>nomMetodo</i>	Texto →	Nombre del método del error

### Descripción

---

El método **Progress SET ON ERROR METHOD** designa un método a ejecutar en caso de un error durante la utilización de las barras de progreso (por ejemplo id desconocido, número de parámetros incorrecto, etc.).

En *nomMetodo*, pase el nombre del método proyecto de la base local a llamar en caso de error. Este método es compartido por todas las ventanas de progreso de la aplicación.

**Nota:** dado que el método **Progress SET ON ERROR METHOD** pertenece a un componente, debe recordar asignar la propiedad "Compartido entre componentes y base local" al método *nomMetodo*; de lo contrario se devuelve un error.

El método *nomMetodo* se llama con tres parámetros::

- \$1 (Entero largo): número del error
- \$2 (Texto): texto del error
- \$3 (Entero largo): identificador único de la barra de progreso

### Ejemplo

---

Este es un ejemplo de un método llamado en caso de error. Este método se declara "compartido" en modo Diseño:

```
C_LONGINT($1)
C_TEXT($2)
C_LONGINT($3)

C_LONGINT($ErrorID)
C_TEXT($ErrorText)
C_LONGINT($ProgressID)

$ErrorID:=$1
$ErrorText:=$2
$ProgressID:=$3
$Error:=""
$Error:=$Error+"Error number: "+String($ErrorID)+Char(Carriage return)
$Error:=$Error+$ErrorText+Char(Carriage return)
$Error:=$Error+"Progress ID: "+String($ProgressID)
ALERT($Error)
```



## Progress SET ON STOP METHOD

Progress SET ON STOP METHOD ( id ; nomMetodo )

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
nomMetodo	Texto	→	Nombre del método de stop

### Descripción

---

El método **Progress SET ON STOP METHOD** designa un método a ejecutar cuando el usuario hace clic en el botón **Stop** de la barra de progreso.

Por defecto, las barras de progreso no tiene botones **Stop**. Si quiere que una barra de progreso tenga uno, debe usar el método **Progress SET BUTTON ENABLED**.

En *id*, pase el identificador único de la barra de progreso, devuelto por el método **Progress New**.

En *methodName*, pase el nombre del método proyecto de la base local a llamar cuando se hace clic en el botón **Stop**. Este método recibe el identificador único de la barra de progreso como su primer parámetro. A continuación, se ejecuta en un nuevo proceso lanzado por el componente.

**Nota:** como el método **Progress SET ON STOP METHOD** pertenece a un componente, debe recordar asignarle la propiedad "Compartido entre componentes y base local" al método *nomMetodo*; de lo contrario se devuelve un error.

## Progress SET PROGRESS

Progress SET PROGRESS ( id ; progreso {; mensaje {; primerPlano}} )

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
progreso	Real	→	Valor de progreso([0...1] o -1)
mensaje	Texto	→	Mensaje de la barra de progreso
primerPlano	Booleano	→	Mostrar la barra de progreso en el primer plano

### Descripción

---

El método **Progress SET PROGRESS** modifica el valor de la barra de progreso junto con la información mostrada en la ventana de progreso. Este método es útil para la actualización de una barra de progreso dentro de un bucle.

En *id*, pase el número de ID único de la barra de progreso, devuelto por el método **Progress New**.

En *progreso*, pase el valor actual de la barra de progreso. Puede pasar un valor Real (entre 0 y 1) ó -1 para especificar una barra de progreso indefinida (también conocido como "Barber shop" bajo Mac OS).

En *mensaje*, pase un texto adicional para mostrar debajo del título principal (Windows) o bajo la barra de progreso (Mac OS). Este parámetro es opcional.


Pase **True** en *primerPlano* cuando quiera forzar el paso de la ventana de progreso al primer plano de la aplicación.

### Ejemplo

---

Actualización de la barra de progreso:

```
$P:=Progress New // creamos una nueva barra
// Efectuar un proceso en un bucle
For($i;1;100)
// ... código del proceso
// Actualizar la barra de progreso
    $r:=$i/100
    Progress SET PROGRESS($P;$r;"More info")
End for
// Eliminación de la barra una vez termina el proceso
PROGRESS QUIT($P)
```





## ⚙️ Progress SET TITLE

Progress SET TITLE ( id ; titulo {; progreso {; mensaje {; primerPlano}} } )

Parámetro	Tipo		Descripción
id	Entero largo	➔	ID de la barra de progreso
titulo	Texto	➔	Título de la barra de progreso
progreso	Real	➔	Valor de progreso ([0...1] o -1)
mensaje	Texto	➔	Mensaje de la barra
primerPlano	Booleano	➔	Mostrar la barra de progreso en el primer plano

### Descripción

---

El método **Progress SET TITLE** define el título de la barra de progreso junto con la información que se muestra en la ventana de progreso.

En *id*, pase el identificador único de la barra de progreso, devuelto por el método **Progress New**.

En *titulo*, pase el texto principal a mostrar en la ventana de progreso.

En *progreso*, pase el valor actual de la barra de progreso (opcional). Puede pasar un valor real (entre 0 y 1) o -1 para definir una barra de progreso indefinida (también conocida como "Barber shop" bajo Mac OS).

En *mensaje*, pase un texto adicional a mostrar bajo el título principal (Windows) o bajo la barra de progreso (Mac OS). Este parámetro es opcional.


Pase **True** en *primerPlano* si desea forzar el paso de la ventana de progreso al primer plano de la aplicación.

### Ejemplo

---

Creación de una ventana de progreso simple:

```
$P:=Progress New  
Progress SET TITLE($P;"Copying folder";-1;"Counting files")
```



## ⚙️ Progress SET WINDOW VISIBLE

Progress SET WINDOW VISIBLE ( visible {; posHor ; posVert {; primerPlano}} )

Parámetro	Tipo		Descripción
visible	Booleano	→	True = Mostrar, False = Ocultar
posHor	Entero largo	→	Coordenada izquierda de la ventana -1 = No cambiar
posVert	Entero largo	→	Coordenada superior de la ventana -1 = No cambiar
primerPlano	Booleano	→	Mostrar barra de progreso en el primer plano

### Descripción

---

El método **Progress SET WINDOW VISIBLE** permite administrar la visualización de la ventana de progresión, si existe.

El parámetro *visible* indica si la ventana debe ser visible o no (por defecto, la ventana es visible). Pase **False** en este parámetro para ocultar la ventana y **True** para mostrarla nuevamente, si existe.

Los parámetros *posHor* y *posVert* modifican la ubicación de la ventana de progreso en la pantalla. En estos parámetros, pase valores que representen el movimiento en píxeles a la derecha (*horPos*) o hacia abajo (*vertPos*) con respecto a las coordenadas de la ventana principal de la aplicación (bajo Windows) o la pantalla (bajo Mac OS).

Para no modificar las coordenadas (por ejemplo, si quiere utilizar el parámetro *primerPlano* sin cambiar la posición de la ventana), pase -1 en cada uno de estos parámetros.

Pase **True** en *primerPlano* si quiere forzar el paso de la ventana de progreso al primer plano de la aplicación.

### Ejemplo 1

---

Ubicar la barra de progreso de 50 píxeles desde la esquina izquierda y 100 píxeles de la esquina superior:

```
Progress SET WINDOW VISIBLE(True;50;100)
```

### Ejemplo 2

---

Ocultar la ventana de progreso:

```
Progress SET WINDOW VISIBLE(False)
```

### Ejemplo 3

---

Mostrar la ventana de progreso y pasarla al primer plano sin modificar su posición:

```
Progress SET WINDOW VISIBLE(True;-1;-1;True)
```

## Progress Stopped

Progress Stopped ( id ) -> Resultado

Parámetro	Tipo		Descripción
id	Entero largo	→	ID de la barra de progreso
Resultado	Booleano	↩	True = el usuario hizo clic en el botón Stop

### Descripción

---

El método **Progress Stopped** devuelve **True** si el usuario hace clic en el botón **Stop** de la barra de progreso designada por el parámetro *id*.

Debe llamar este método para probar si el usuario hizo clic en el botón **Stop**. El botón no dispara un evento.

### Ejemplo

---

Ejemplo de barra de progreso en un bucle:

```
$ProgressID:=Progress New // creación de una nueva barra
// La barra de progreso debe tener un botón Stop
Progress SET BUTTON ENABLED($ProgressID;True)
For($i;1;100)
// Mientras el progreso no se detenga...
  If(Not(Progress Stopped($ProgressID)))
    Progress SET TITLE($ProgressID;"Test progress #"+String($ProgressID))
    Progress SET PROGRESS($ProgressID;$i/100)
    Progress SET MESSAGE($ProgressID;String(100*$i/100)+" %")
    (...)
  Else // El usuario hace clic en Stop
    $i:=100 // Salimos del bucle
  End if
End for
// Cierre final de la barra (el botón Stop no hace nada)
Progress QUIT($ProgressID)
```